Formulare schneller öffnen
Versionen: Access 2003, 2002/XP, 2000 und 97

Wenn das Öffnen eines Formulars einige Augenblicke dauert, wissen Sie als Entwickler was im Hintergrund abläuft. Sie kennen die Abfrage, die Access zunächst ausführen muss oder die Datenmengen der Tabelle, auf der das Formular basiert. Anwender haben aber meistens wenig Verständnis dafür und daher zeigen wir Ihnen in dieser Woche einen kleinen Trick, mit dessen Hilfe Sie Formulare auf den ersten Blick ohne Wartezeit sofort zur Verfügung stellen können. Tatsächlich wird der Ladevorgang nur verlagert, aber Sie werden sehen: die Umstellung wirkt Wunder!

Das Konzept: In einer Tabelle legen Sie fest, welche Formulare der Datenbank beim Start direkt geöffnet werden sollen. Über eine zusätzliche Einstellung können Sie angeben, in welchem Modus ein Formular geöffnet werden soll. Über "AutoExec" wird dann beim Start der Datenbank eine Funktion aufgerufen, die die gewünschten Formulare öffnet. Wechseln Sie später zur Datenbank, können Sie direkt mit den benötigten Formularen arbeiten, ohne lange Wartezeiten in Kauf nehmen zu müssen.

Um die Lösung in Ihre Datenbank einzubauen, gehen Sie wie folgt vor:

· Legen Sie eine Tabelle mit folgenden Feldern an:

FormularName: Datentyp "Text", Feldgröße: 100 Zeichen
AnzeigeModus: Datentyp "Zahl", Feldgröße: "Byte"
· Setzen Sie den Primärschlüssel auf das Feld "FormularName".

· Wechseln Sie auf die Registerkarte "Nachschlagen" und richten Sie folgendes Steuerelement ein:

Steuerelement anzeigen: Kombinationsfeld
Herkunftstyp: Werteliste 
Datensatzherkunft: Versteckt;0;Minimiert;1;Normal;2 
Gebundene Spalte: 2 
Anzahl Spalten: 2 
Spaltenbreiten: 3cm;0cm
· Speichern Sie die Tabelle unter dem Namen "AutoLaden_Formulare".

· Verlassen Sie den Tabellenentwurf.

· Wechseln Sie in die VBA-Entwicklungsumgebung.

· Wählen Sie das Menü Extras-Verweise an, lokalisieren Sie den Eintrag "Microsoft DAO 3.60 Object Library" (Access 2003, 2002/XP und 2000) beziehungsweise "Microsoft DAO 3.51 Object Library" (Access 97), aktivieren Sie das Kontrollkästchen vor dem Eintrag und klicken Sie dann auf Ok.

· Legen Sie ein neues Modul an und geben Sie die folgende Funktion ein:

Public Function AutoLaden()
  Dim db As DAO.Database
  Dim rs As DAO.Recordset
  Dim strForm As String
  Dim bModus As Byte
  Dim Anz As Long, I As Long

  On Error Resume Next
  Set db = CurrentDb()
  Set rs = db.OpenRecordset("AutoLaden_Formulare")
  Err = 0
  rs.MoveLast
  If Err = 0 Then
    Anz = rs.RecordCount
    rs.MoveFirst
    For I = 1 To Anz
      strForm = rs("FormularName")
      bModus = rs("AnzeigeModus")
      Select Case bModus
        Case 0
          DoCmd.OpenForm strForm, , , , , acHidden
        Case 1
          DoCmd.OpenForm strForm, , , , , acIcon
        Case 2
         DoCmd.OpenForm strForm, , , , , acWindowNormal
      End Select
      rs.MoveNext
    Next I
  End If 'Err= 0...
  rs.Close

  Set rs = Nothing
  Set db = Nothing
  DoEvents

End Function

Im ersten Schritt öffnen wir hier die Tabelle "AutoLaden_Formulare" als Recordset und ermitteln zur Begrenzung der folgenden Schleife die Anzahl vorhandener Datensätze. Kommt es dabei zu einem Fehler, ist die Tabelle leer beziehungsweise es sollen keine Formulare angezeigt werden. Die entsprechenden Anweisungen werden also nur bei "Err = 0" ausgeführt. In einer Schleife gehen wir dann alle Datensätze durch, lesen den Formularnamen und den Anzeigemodus aus und rufen dann je nach Anzeigemodus "DoCmd.OpenForm" mit einem der Parameter "acHidden", "acIcon" oder "acWindowNormal" auf.

· Speichern Sie das Modul unter dem Namen "modAutoLaden" und verlassen Sie die VBA-Entwicklungsumgebung wieder.

Damit die Formulare beim Start der Datenbank automatisch geöffnet und initialisiert werden, nutzen wir das "AutoExec"-Makro für den Aufruf der Funktion "AutoLaden()". Wenn in Ihrer Datenbank bereits ein "AutoExec"-Makro vorhanden ist, fügen Sie am Anfang eine weitere Aktion "AusführenCode" ein und geben für den Parameter "Funktionsname" den Ausdruck "AutoLaden()" ein. Ist in Ihrer Datenbank kein "AutoExec"-Makro vorhanden, legen Sie es zunächst im Bereich "Makros" des Datenbankfensters per Schaltfläche Neu an, richten dann die oben erwähnte Aktion ein und speichern es unter dem Namen "AutoExec".

Über die Tabelle "AutoLaden_Formulare" legen Sie fest, welche Formulare in welchem Modus automatisch beim Start geladen werden sollen. Dazu geben Sie den Namen des Formulars, so wie er im Datenbankfenster angezeigt wird, im Feld "FormularName" ein. Über das Feld "AnzeigeModus", für das ein Nachschlage-Kombinationsfeld eingerichtet ist, wählen Sie dann den gewünschten Modus "Versteckt", "Minimiert" oder "Normal" aus. Beim Modus "Versteckt" wird das Formular geladen, bleibt aber komplett unsichtbar. Es ist aber, wenn Sie es benötigen, bereits komplett initialisiert und einsatzbereit. Sie können es entweder über das Menü Fenster-Einblenden anzeigen lassen oder es einfach per Doppelklick im Datenbankfenster öffnen - Access macht dann die bereits im Speicher befindliche, initialisierte Kopie sichtbar. Beim Modus "Minimiert" wird das Formular als Symbol am unteren Fensterrand von Access abgelegt. Sie können sich auf diesem Wege beispielsweise Verwaltungsformulare, die nicht sofort benötigt werden, vorab als Symbol zurechtlegen. Im Modus "Normal" wird das Formular wie gewohnt zur direkten Bearbeitung angezeigt. In Kombinationen aus "Normal" und "Minimiert" können Sie also zum Beispiel ein Formular zur Terminverwaltung für die sofortige Prüfung anstehender Termine und Aufgaben normal öffnen und Artikelinfos und Kundenadressen für den späteren Einsatz minimiert ablegen.

